Dateien und Verzeichnisse auf Existenz prüfen
Versionen: Access 2003, 2002/XP, 2000 und 97

 FRAGE:  Wir möchten gerne folgende Routine nutzen, um Datensicherungskopien von Datenbanken anzulegen, sofern die Datenbank nicht im Zielverzeichnis vorhanden ist:

strPath = "Z:\Test\"
strDatsi = "Z:\Datsi\"
strFName = Dir$(strPath & "*.mdb")
While strFName<> ""
  DoEvents
  If Dir$(strDatsi & strFName) = "" Then
    FileCopy strPath & strFName, strDatsi & strFName
  End if
  strFName = Dir$()
Wend

Die Routine funktioniert immer nur bis zur ersten Datenbank, die nicht im Zielverzeichnis vorhanden ist, danach wird das Nichtvorhandensein weiterer, ggf. zu sichernder Datenbanken nicht mehr erkannt. Was ist daran falsch?
U. Roth

 ANTWORT:  Sie können "Dir$()" nicht verschachtelt einsetzen, da der Aufruf ohne Parameter (strFName = Dir$()) sich immer am letzten Aufruf mit Parameter (If Dir$(strDatsi & strFName) = "" Then...) orientiert und so das ursprüngliche Ziel (strFName = Dir$(strPath & "*.mdb")) "vergisst". Abhilfe können Sie schaffen, indem Sie für die Abfrage auf Vorhandensein einer Datei (oder eines Verzeichnisses) eine Alternative einsetzen: Die Microsoft Scripting Runtime Library. Diese Systemerweiterung stellt Objekte bereit, über die Sie unter anderem auf das Dateisystem zugreifen und so Dateien und Verzeichnisse auf Vorhandensein prüfen können:

1. Öffnen Sie ein vorhandenes oder ein neues Modul im Entwurfsmodus.

2. Wählen Sie das Menü Extras-Verweise an, setzen Sie eine Referenz auf "Microsoft Scripting Runtime" und übernehmen Sie die Änderungen mit einem Klick auf Ok.

3. Geben Sie im Modul die folgende Funktion ein:

Function CheckFileOrDirExists(strName As String) _
As Boolean

  Dim fso As Scripting.FileSystemObject

  Set fso = CreateObject("Scripting.FileSystemObject")
  If Right$(strName, 1) = "\" Then 'Verzeichnis
    CheckFileOrDirExists = fso.FolderExists(strName)
  Else
    CheckFileOrDirExists = fso.FileExists(strName)
  End If
End Function
· Speichern Sie die Änderungen.

In Zukunft können Sie diese Funktion nun ohne Rücksicht auf einen eventuell parallel stattfindenden Einsatz der "Dir$()"- Funktion einsetzen, um das Vorhandensein von Dateien und Verzeichnissen beispielsweise wie folgt zu prüfen:

strPath = "Z:\Test\"
strDatsi = "Z:\Datsi\"
strFName = Dir$(strPath & "*.mdb")
While strFName<> ""
  DoEvents
  If Not CheckFileOrDirExists(strDatsi & strFName) Then
    FileCopy strPath & strFName, strDatsi & strFName
  End if
  strFName = Dir$()
Wend

Sie übergeben der Funktion also entweder den Namen der zu prüfenden Datei wahlweise mit Laufwerk und/oder Pfad oder den Pfad des zu prüfenden Verzeichnisses mit abschließendem Backslash. An einem eventuell vorhandenen abschließenden Backslash erkennt die Funktion zunächst, ob ein Verzeichnis oder eine Datei zu prüfen ist. Abhängig davon wird entweder die Funktion "FolderExists()" oder die Funktion "FileExitsts()" des FileSystemObjects aufgerufen.

